

团 体 标 准

JH/CIE 170-2021

区块链 智能合约 合同文本置标语言(CTML)

Blockchain—Smart contract—Contract text markup language (CTML)

（征求意见稿）

（本稿完成时间：2021 年 9 月 25 日）

在提交反馈意见时，请将您知道的相关专利连同支持性文件一并附上。

2020- XX -XX 发布

2020- XX -XX 实施

中国电子学会 发布

目 次

前言 II

引言 III

1 范围 1

2 规范性引用文件 1

3 术语和定义 1

4 缩略语 3

5 符号和关键词 3

 5.1 标注符号 3

 5.2 关键词 4

6 拼写规则 4

7 CTML 置标体系..... 5

 7.1 原则 5

 7.2 CTML 记法..... 5

 7.3 CTML 使用流程..... 6

 7.4 CTML 合同类别..... 6

 7.5 CTML 置标要求..... 7

8 层级标注结构 7

 8.1 概述 7

 8.2 法律要素标注 7

 8.3 法律属性标注 8

 8.4 法律成分标注 8

 8.5 域标注 9

9 要素构成 9

10 要素表述 11

 10.1 合约框架 11

 10.2 合同标题 11

 10.3 当事人描述 12

 10.4 标的 12

 10.5 合约条款 14

 10.6 资产操作 17

 10.7 资产表达式 19

 10.8 时间表达式 19

 10.9 附加信息 22

 10.10 合约订立 22

附 录 A （资料性） 法律文本合同及标注后 CTML 合同示例 24

 A.1 法律文本合同 24

 A.2 标注后 CTML 合同 25

附 录 B （资料性） EMD 的交互数据属性与示例.....	29
B.1 EMD 的交互数据属性.....	29
B.2 EMD 示例.....	29
附 录 C （资料性） CTML 到 SLCL 转化关系表.....	31
参考文献	34
图 1 合同转换关系示意图.....	III
图 2 面向法律文本合同的智能合约生成和执行方法示意图.....	7
图 3 CTML 构成要素关系图	10
表 1 关键词及其含义.....	4
表 2 交换数据表示例.....	30
表 3 CTML 到 SLCL 转化关系表	31

前 言

本文件按照GB/T 1.1—2020《标准化工作导则 第1部分：标准化文件的结构和起草规则》的规定起草。

本文件由北京科技大学提出。

本文件由中国电子学会区块链分会技术归口。

本文件起草单位：北京科技大学、北京大学、北京市经济和信息化局、北京市社会公信建设促进会、天津蛟图信息科技有限责任公司、云南省区块链应用技术重点实验室、贝壳找房（北京）科技有限公司、浙江蚂蚁小微金融服务集团有限公司、中国电子技术标准化研究院、中国移动通信集团有限公司。

本文件主要起草人：朱岩、王迪、范雨晴、林鸿杰、孙贻滋、郭倩、王晟典、欧阳欢、洪京一、姚前、蒋国庆、李冀宁、刘国伟、伊然、冯荣权、胡凯、韩心慧、英伟、张韬、林映春、唐晓丹、董宁、昌文婷。

引 言

近年来随着区块链技术的迅速发展和广泛应用，建立在区块链基础上的智能合约技术也日益成熟，被称为新一代区块链的核心技术。然而，智能合约目前尚不能被视为一种法律合同，其编写的智能合约程序也缺少法律认可和相应的法律效力，因此智能法律合约SLC在《区块链智能合约形式化表达》（T/CIE 095—2020）标准中被提出，它为法律文本合同与智能合约代码之间建立了转化的桥梁。目前法律文本合同到智能法律合约的转化缺乏规范化和准确性，即智能法律合约的编写依赖于程序人员主观对法律文本合同的理解，难以准确体现原始法律文本合同的意思表示，因而依然无法明确智能法律合约与原始法律文本合同具有相同法律地位。

本文件属于《区块链 智能合约》标准系列，通过提供一种合同文本置标语言CTML，使用语义标记和数源标记对法律文本合同中法律相关元素进行标注，避免自然语言可能产生的二义性。同时，构造交换标记数据表EMD实现合同语义与用户数据分离，支持当事人之间的合同协商与合约执行过程中数据交互。此外，使用层级标注结构建立法律文本合同与智能法律合约之间的准确映射关系与相同意思表示。

如图1所示，以自然语言为载体的法律文本合同通过合同文本置标语言标注后生成CTML合同，再由转化为智能法律合约语言撰写的智能法律合约，最后经编译后生成智能合约可执行代码，从而实现了由法律文本合同到智能合约代码生成的完整且规范化流程，保证转化后的程序与所标记的法律合同具有同等法律效力。



图1 合同转换关系示意图

本文件采用合同文本置标语言对文本合同中法律要素进行准确描述和解释，消除自然语言表述可能产生的二义性，使得不同领域人员可对合同内容进行有效识别和无偏差理解，实现文本合同到智能法律合约的规范化转化和自动化生成，有利于对法律合同的进一步形式化分析与智能化应用。

本文件的发布机构提请注意，声明符合本文件时，可能涉及到第5、7章与合同文本置标语言语法相关的专利的使用。

本文件的发布机构对于该专利的真实性、有效性和范围无任何立场。

该专利持有人已向本文件的发布机构承诺，他愿意同任何申请人在合理且无歧视的条款和条件下，就专利授权许可进行谈判。该专利持有人的声明已在本文件的发布机构备案。相关信息可以通过以下联系方式获得：

专利：202110162638.5 一种基于合同文本标记语言的法律合同交互式标注方法

专利持有人姓名：北京科技大学

地址：北京市海淀区学院路30号

请注意除上述专利外，本文件的某些内容仍可能涉及专利。本文件的发布机构不承担识别专利的责任。

区块链 智能合约 合同文本置标语言(CTML)

1 范围

本文件规定了合同文本置标语言的结构和语法，界定了有关术语和定义。

注：本文提供的合同文本置标语言示例及转化后智能法律合约示例，见附录 A。

本文件适用于通用智能合约设计、开发和使用，也可为区块链厂商构建智能合约平台提供参考。

2 规范性引用文件

下列文件中的内容通过文中的规范性引用而构成本文件必不可少的条款。其中，注日期的引用文件，仅该日期对应的版本适用于本文件；不注日期的引用文件，其最新版本(包括所有的修改单)适用于本文件。

GB/T 14814—1993 信息处理 文本和办公系统 标准通用置标语言(SGML)

T/CIE 095—2020 区块链 智能合约 形式化表达

3 术语和定义

下列术语和定义适用于本文件。

3.1

区块链 blockchain

使用密码链接将共识确认过的区块组织按顺序追加形成的分布式账本。

[来源：ISO 22739:2020, 3.6]

3.2

智能合约 smart contract

部署在区块链上、在满足预定条件时可自动执行并存证的计算机程序。

3.3

智能法律合约 smart legal contract

一种含有合同构成要素、涵盖合同缔约方依据要约和承诺达成履行约定的计算机程序。

3.4

智能合约语言 smart contract language

用于定义智能合约、包含词汇和语法规则的形式化表达规范。

3.5

智能法律合约语言 smart legal contract language

可实现符合法律的智能合约的编程语言。

3.6

文本合同 textual contract

采用自然语言撰写，用以当事人记载合同内容的书面文件。

3.7

CTML 合同 CTML contract

符合本文件的CTML实例文档。

3.8

法律要素 law factor

构成合同内容所必要的基本因素或元素，一般包括标题、当事人、资产、条款、合同订立及附加信息等。

注：在不引起混淆的情况下，本文件中简称为“要素”。

3.9

法律属性 law property

构成法律要素的基本性质，一般包括资产的性质与权属，条款中的当事人行为、前置条件、伴随条件、后置条件、违反、争议、机构，合同订立中的签名等。

注：在不引起混淆的情况下，本文件中简称为“属性”。

3.10

法律成分 law component

法律属性表达所涉及到的限定信息，包括时间表达中的行为时间、时间谓词、边界谓词、资产操作中的存入动作、取回动作、转移动作、资产表达式等。

注：在不引起混淆的情况下，本文件中简称为“成分”。

3.11

域 field

用于描述法律要素、属性、成分的特性及其取值。

3.12

特征 attribute

依附于法律要素、属性、成分等实体中，以#为前缀表示，用于描述合同中有别于其他的实体特点。

3.13

合同文本置标语言 contract text markup language

将合同文本以及文本相关信息结合起来，展现出关于文档结构、法律特征、词汇意义和意思表示，以及数据处理细节的计算机可处理文字编码。

3.13.1

语义标记 semantic markup

根据合同条款所蕴含的意义，使用特定符号对文本合同中要素、属性、成分等进行辨识。

注：在不引起混淆的情况下，本文件中也称为可嵌套标记或复杂标记。

3.13.2

数源标记 metadata markup

用于对合同中基本且不可分割的未确定内容进行辨识，可被当事人进行宣称、填入或选择。

注：在不引起混淆的情况下，本文件中也称为不可嵌套标记或简单标记。

3.14

（CTML）嵌套结构 nested structure (of CTML)

通过包含关系描述要素、属性、成分间的层次结构。

3.15

客户 clients

使用网络终端对CTML合同进行协商、签订和履行的合同当事人。

3.16

合约服务器 contract server

是一种计算机系统，用于对CTML合同与EMD进行管理，并可转化生成SLC程序及智能合约可执行代码。

3.17

交换标记数据表 exchange markup datasheet

以“键-值对”形式描述 CTML 合同中数源标记的数据表，用于对客户操作数源标记的交互过程进行确认、特指或限制。

4 缩略语

下列缩略语适用于本文件

CTML	合同文本置标语言（Contract Text Markup Language）
SLC	智能法律合约（Smart Legal Contract）
SLCL	智能法律合约语言（Smart Legal Contract Language）
EMD	交换标记数据表（Exchange Markup Datasheet）
LFM	法律要素标注（Law Factor Marking）
LPM	法律属性标注（Law Property Marking）
LCM	法律成分标注（Law Component Marking）

5 符号和关键词

5.1 标注符号

下面符号适用于本文件：

@	标识前缀
::=	表示定义，即“被定义为”

	同级元素或
[]	关键词任选，可以为空
#	可选特征指示符
%	类型指示符
+	零或多项
.	层级关系连接符
<</>> <</>>	语义标记，前部表示标记开始，后部表示标记结束
<{ }>	数源标记

5.2 关键词

本文件中所用关键词及其对应的含义如表 1 所示。

表 1 关键词及其含义

关键词	含义
factor, property, component, field	法律要素、法律属性、法律成分、域
all, some, this	任意、存在、当前的限制冠词
can, must, cannot	权利限制，应当限制，禁止限制
true, false	真与假的布尔值
month, day, year, hour, minute, second, now	月、日、年、小时、分钟、秒、当前的时间符号
String, Money, Date, Integer, Float, Boolean, Time	表示字符串、货币、日期、整数、浮点数、布尔、时间的数据类型（dataType）
Right, useRight, ownershipRight, possessRight, usufruct, disposeRight	表示专有权属、使用权、所有权、占有权、收益权、处分权的权属类型（rightType）
title, party, group party, asset, genTerm, breTerm, arbiTerm, addition, conclusion	标题、个体当事人、群体当事人、标的、一般条款、违约条款、仲裁条款、附加信息、合约订立等法律要素
info, right, action, preCondition, adjCondition, postCondition, against, controversy, institution, signature	资产属性信息、资产权属信息、行为、前置条件、伴随条件、后置条件、违反、争议、机构、签名等法律属性
actionTime, timePredicate, rangePredicate, transfer, withdraw, deposit, assetExpression	行为时间、时间谓词、边界谓词、转移动作、取回动作、存入动作、资产表达式等法律成分
quantity, attribute, serialNumber, terms, party, duty, action, limit, targetTime, judge, prep, baseTime, within, boundary, condition, target, amount, assetRight, assetTarget	阈值、特征、序号、子条款、当事人、义务、行为、限制冠词、目标时间、判断词、时间方向介词、基准时间、时间范围、边界判定、价值要求、转移目标、资产数量、资产权属、特定资产等法律特征
term, breach term, arbitration term, Contract conclusions, signature of party, additions, serial number, institution, when, while, where	一般条款、违约条款、仲裁条款、合约订立、当事人签名、附加信息、序列号、机构、执行、伴随、执行后等 SLCL 关键词

6 拼写规则

置标采用本文件所提供的英文置标语法。

CTML 标记中单词区分字母大小写，可完整使用，在名称中不使用缩略语，以保证语义清晰，提高可读性。

CTML标记中要素、属性、特征等需要客户提供唯一性标识，该标识只能使用字母，不使用任何特殊符号。

CTML 采用表达式规范合约内容。表达式中使用的符号包括：

- 常量符号，包括：数字、字符串、**true**、**false**；
- 时间符号，包括：**month**、**day**、**year**、**hour**、**minute**、**second**、**now**；
- 类型符号，包括：**String**、**Money**、**Date**、**Integer**、**Float**、**Boolean**、**Right**、**Time**。

7 CTML 置标体系

7.1 原则

本章定义合同文本置标语言的总体结构，该结构用于合同生成和履行等过程对合同信息的描述。

CTML结构的建立原则是：

- a) 以智能合约领域业务对合同内容的需求分析为基础；
- b) 独立于合同样式的表现；
- c) 独立于特定合同类型；
- d) 建立与SLCL的对应关系。

本文件对文本合同内容本身、条款不同内容之间关系以及条款内容的各类文档结构、特征、词汇的法律意义进行建模，并定义这些内容的描述方法和数据处理方法。

7.2 CTML 记法

7.2.1 语义标记

语义标记具有指示性作用并可提供法律上的意思表示。置标信息置于双尖括号内部并将其置于被标注文本外侧，即<<置标信息>>文本<</置标信息>>。

语义标记采用如下格式：

```
语义标记::=
    <<要素|属性|成分|域 参数列表>>
    文本
    <</要素|属性|成分|域>>
```

对应置标语法为

```
semanticMarkup::=
    <<factor|property|component|field parameterList>>
    text
    <</factor|property|component|field>>
```

其中，

factor	要素保留字，
property	属性保留字，
<i>component</i>	具体法律成分保留字，
field	域保留字，
parameterList	参数列表，依据标记种类具有不同定义，
text	被标注文本。

7.2.2 数源标记

数源标记的对象是 CTML 合同中可交互数据。

数源标记采用如下格式：

数源标记::=

<{[要素标识]@交互数据 [%类型] (#选择方式=备选数据)+}>

对应英文表示为

metadataMarkup::=

<{[factorID]@exchangedData [%type] (#option=value)+}>

其中，

factorID	交互数据所属层次中最外层的要素标识，
exchangedData	该交互数据的唯一标识，
type	交互数据的数据类型，
option	交互数据的类型，
value	合同文本在此处的备选数据或取值。

注 1：option 可包括

- singleOption（单选）表示可接受用户在选择列表中唯一选定；
- multiOption（多选）表示可接受用户在选择列表中一或多个选定；
- import（外部输入）表示可接收用户传入数据；
- trigger（触发）表示可接收外部事件；
- allocate（分配）表示可接收用户定义的复杂类型数据。

value 在无异议的情况下，本文档中 value 皆表示上述同等含义。

type 分为数据类型和权属类型，在无异议的情况下，本文档中 type 皆表示上述同等含义。

注 2：嵌套结构可以通过层级关系连接符加以表示，数源标记可在文本中任何地方被使用。

示例 1：

合同文本：

房屋的价格

标注后文本：

<{asset@House.price %Money}>

示例 2：

合同文本：

房屋使用权获取方式可选方式一为划拨

标注后文本：

<{@accessMethod %Right #singleOption={allocation}}>

7.3 CTML 使用流程

CTML支持文本合同到智能合约代码生成的完整且规范化流程，如图2所示，流程如下：

- a) 支持基于文本合同的智能合约开发与部署，步骤包括：
 - 1) 文本合同采用合同文本置标语言标注后生成CTML合同和EMD；
 - 2) 通过CTML到SLCL的词汇映射和转化规则生成SLC程序；
 - 3) SLC程序经编译后与EMD链接生成智能合约可执行代码；
 - 4) 智能合约代码部署至智能合约平台，实现合约部署。
- b) CTML合同、EMD、SLC程序整合至合约服务器，宜于合同相关方与合约服务器间实时交互及合约服务器与智能合约平台间通信，实现合同协商、订立与执行。

7.4 CTML 合同类别

由 CTML 标注的文本合同按照标记程度分为：

- 基本标注合同，仅采用数源标记的法律合同，宜用于法律合同与智能合约之间的数据交互；
- 语义标注合同，仅采用语义标记的法律合同，宜用于被标注合同到智能法律合约的转化；
- 完全标注合同，采用语义标记和数源标记的法律合同，既可用于被标注合同到智能法律合约的转化，又可用于法律合同与智能合约之间的数据交互。




图 2 面向法律文本合同的智能合约生成和执行方法示意图

7.5 CTML 置标要求

CTML 置标过程应以促成和确保 CTML 合同与原始文本合同具有相同法律效力为目标，基本要求包括：忠实于原文的内容，将原文的内容充分表达出来，无任意增减删略或歪曲背离，消除文法的二义性；合理使用 CTML 语法规则，力求简明通顺、易于理解。

8 层级标注结构

8.1 概述

CTML 合同中语义标记采用层级语义标注结构，按包含关系分为法律要素标注、法律属性标注、法律成分标注以及辅助性标注。

8.2 法律要素标注

LFM 属于一级标注功能，用于提取和识别文本合同中法律要素，并形成要素表达式。语法如下：

要素表达式::=
 <<要素 要素类别@要素标识 (#特征=特征值)+>>
 文本
 <</要素>>

对应英文格式为：

factorExpression ::=
 <<factor factorSet@factorID (#attribute=value)+>>
 text

<</factor>>

其中,

factor 要素保留字,
factorID 要素标识,
attribute 当前要素所具有的法律特征,
value 特征的取值,
factorSet 当前要素具体的要素类别, 表示如下:
 要素类别::=
 {标题, 当事人|群体, 资产, 一般条款|违约条款|仲裁条款,
 合同订立, 附加信息}

对应英文表示如下:

factorSet ::=
 {**title, party|group, asset, genTerm|breTerm|arbiTerm,**
conclusion, addition}

8.3 法律属性标注

LPM 属于二级标注功能, 用于提取和识别要素中的法律属性, 并形成属性表达式。语法如下:

属性表达式::=
 <<属性 [要素标识.]属性类别[@属性标识] (#特征=特征值)+>>
 文本
 <</属性>>

对应英文格式为:

propertyExpression ::=
 <<**property** [factorID.]*propertySet*[@propertyID] (#attribute=value)+>>
 text
 <</**property**>>

其中,

property 属性保留字,
propertyID 属性标识,
propertySet 当前属性具体的法律属性, 表示如下:
 属性类别::=
 {信息, 权属, 行为, 前置条件, 伴随条件, 后置条件,
 违反, 争议, 机构, 签名},

对应英文表示为

propertySet ::=
 {**info, right, action, preCondition, adjCondition, postcondition,**
against, controversy, institution, signature}

注 1: 如果 LPM 被嵌套到一级标注中, 则可对要素标识。(即 factorID.) 予以省略。

注 2: 当要素标识对应实体中属性的个数有且只有一次时, 则可对@属性标识(即@propertyID)予以省略。

8.4 法律成分标注

LCM 属于三级标注功能, 用于嵌套式提取与识别特定要素、属性及成分中的法律成分, 并形成成分表达式, 其中, LCM 内可嵌套其他 LCM。语法如下:

成分表达式::=
 <<成分 (#特征=特征值)+>>
 文本
 <</成分>>

对应英文格式:

componentExpression ::=
 <<component (#attribute=value)+>>
 text
 <</component >>

其中,

component 成分集合中指定的法律成分, 表示如下:

成分集合::=
 {行为时间, 时间谓词, 边界谓词,
 存入动作|取回动作|转移动作, 资产表达式},

对应英文表示为

componentSet ::=
 {actionTime, timePredicate, rangePredicate,
 deposit|withdraw|transfer, assetExpression}

8.5 域标注

域标注属于一种辅助性的标注功能, 用于提取和识别要素、属性、成分中的域信息, 并形成域表达式。语法如下:

域表达式 ::=
 <<域 [要素标识.属性集]@域标识 [%类型] [#域值=取值]>>
 文本
 <</域>>

对应英文格式:

fieldExpression ::=
 <<field [factorID.propertySet]@fieldID [%type] [#quantity=value]>>
 text
 <</field>>

其中,

quantity 域值保留字, 等号右侧是当前域的取值。

示例:

<<property info>>房屋状况
 <<field @location>>房屋坐落<{House@location}> <</field>>
 <<field @area>>建筑面积<{House@area %Interger}> <</field>>
 <</property>>

9 要素构成

CTML按照层级标注结构实施文本合同中法律要素的标注。标注元素关系图如图3所示, 具体要素有以下六种:




图 3 CTML 构成要素关系图

- 合约名称，以title形式加以描述；
- 当事人，包括个体当事人（party）和群体当事人（group party）；
- 标的，用于声明资产所需信息，包括当前资产属性信息（info）与权属信息（right），两者均属于资产要素下的法律属性；
- 合约条款，用于当事人权利和义务的宣称，包括一般条款（genTerm）、违约条款（breTerm）和仲裁条款（arbiTerm），属于法律要素。说明如下：
 - 一般条款和违约条款均以行为（action）、前置条件（preCondition）、伴随条件（adjCondition）和后置条件（postCondition）等法律属性对条款构成进行标注，

- a) 前置条件和后置条件涉及时间信息的提取和表示，被提取的时间信息按照类别可分为行为时间（**actionTime**）、时间谓词（**timePredicate**）、边界谓词（**rangePredicate**）三种，属于法律成分，
- b) 伴随条件可指定的资产操作方式包括存入（**deposit**）、取回（**withdraw**）和转移（**transfer**），并使用资产表达式（**assetExpression**）对资产操作进行调用，上述元素均属于法律成分，
- c) 违约条款使用违反（**against**）属性对所针对条款进行标注。

——仲裁条款应依据合同陈述的事实对其规定的争议（**controversy**）和机构（**institution**）进行标注；

——合同订立，应包含当事人的签名（**signature**），属于法律属性；

——附加信息，根据需要可采用**addition**形式加以描述。

10 要素表述

10.1 合约框架

CTML 合同包括文档类型说明、文档内容两部分。文档类型说明包含语言类型、具体文档类型、对应的 EMD。

示例：

```
<!DOCTYPE ctml>
<ctml lang=zh_cn, link="exchangesheet.emd">
  <<factor title@printerDeal>> 打印机购买合同 <</factor>>
  hello world!
</ctml>
```

10.2 合同标题

合同标题用于描述法律合同文件相关信息及其标注，语法如下：

```
<<factor title@titleID [#serialNumber=value]>>
    text
<</factor>>
```

其中，

title	标题保留字，
titleID	合约标识，
serialNumber	序号保留字，可缺省，
value	合约序号的具体取值。

示例：

合同文本：

房屋买卖合同

标注后文本：

```
<<factor title@purchase #serialNumber =0x827198...ab193>>
    房屋买卖合同
<</factor>>
```

注：该合约序号为 0x827198...ab193。

10.3 当事人描述

当事人标记用于对当事人相关信息进行宣称，包括当事人唯一性的身份信息、联系方式等。当事人又分为个体当事人和群体当事人，标记通过当事人标识对所有已声明当事人进行区分。语法如下：

——个体当事人标记：

```
<<factor party@partyID>>
    text
<</factor>>
```

——群体当事人标记：

```
<<factor group (party)?@groupID>>
    text
<</factor>>
```

其中，

party	个体当事人保留字，
partyID	个体当事人标识，
group party	群体当事人保留字， party 可省略，
groupID	群体当事人标识。

示例：

合同文本：

拍卖方举办的拍卖会上，竞标人竞标拍卖物。

标注后文本：

```
<<factor party@Seller>>
    拍卖方
<</factor>>举办的拍卖会上，
<<factor group@Voters>>
    竞标人
<</factor>>竞标拍卖物。
```

注：群组当事人可表示为动态加入的当事人列表。

10.4 标的

标的标记用于描述当事人权利和义务共同指向的对象以及对象的权属关系，其中，对象被称为资产，一般分为物、行为、智力成果等。标记通过资产标识对已声明资产进行区分。语法如下：

```
<<factor asset@assetID>>
    text
<</factor>>
```

其中，

asset	资产保留字，
assetID	资产标识。

声明资产时需标注信息包括与合约相关资产属性信息（info）与资产权属信息（right），两者均属于资产要素下的法律属性。语法如下：

——资产属性信息（info）：

```
<<property [assetID.]info>>
    text
<</property>>
```

——资产权属信息（right）：

```
<<property [assetID.]right>>
    text
<</property>>
```

其中，

info 资产属性信息保留字，
right 资产权属信息保留字，
assetID 指定资产标识。

info 与 right 中的 text 可进一步使用域标记对其拥有的特性及取值进行标定。right 在标记类型时应使用权属类型，其中，权属类型定义如下：

权属类型::=

{专有权属，使用权，所有权，占有权，收益权，处分权}

对应英文表示如下：

rightType ::=

{**Right, useRight, ownershipRight, possessRight, usufruct, disposeRight**}

注：专有权属是一种特定属性下用户自定义的权属。

——如果标的物属性采用非集中方式进行描述，语法如下：

```
<<field assetID.info@fieldID %dataType>>
    text
<</field>>
```

——如果标的物权属采用非集中方式进行描述，语法如下：

```
<<field assetID.right@fieldID %rightType>>
    text
<</field>>
```

示例 1：

合同文本：

甲方将自有的房屋出租给乙方。

标注后文本：

甲方将自有的<<factor asset@House>>房屋<</factor>>给乙方

示例 2：

合同文本：

房屋状况填写：房屋坐落位置为___，幢号为___，室号为___，建筑面积为___平方。

标注后文本：

```
<<property House.info>>
```

房屋状况填写：房屋坐落位置为___，幢号为___，室号为___，建筑面积为___平方。

```
<</property>>
```

注：资产属性信息 info 的标记文本中，使用域标记和数源标记对拥有的特性及取值进行逐一详细标注。完整标注后文本：

```
<<property House.info>>房屋状况填写：
```

```
<<field @location>>房屋坐落位置为<{House@location}><</field>>
```

```
<<field @blockNumber>>幢号为<{House@blockNumber}><</field>>
```

```
<<field @roomNumber>>室号为<{House@roomNumber}><</field>>
```

```
<<field @area %Integer>>建筑面积为<{House@area %Integer}><</field>>平方
```

```
<</property>>
```

示例 3:

合同文本:

土地使用权取得方式“√”: 1.出让 2.划拨。

标注后文本:

```
<<property House.right>>
```

土地使用权取得方式 “√”: 1.出让 2.划拨。

```
<</property>>
```

注: 资产属性信息 **right** 的标记文本中, 使用域标记和数源标记对拥有的特性及取值进行逐一详细标注。完整标注后文本:

```
<<property House.right>>
```

```
<<field @houseLandUse %Right>>土地使用权取得方式“√”:
```

```
1.出让 <{@accessMethod %Right #singleOption={transferred}}>
```

```
2.划拨 <{@accessMethod %Right #singleOption={allocation}}>
```

```
<</field>>
```

```
<</property>>
```

10.5 合约条款

10.5.1 一般条款

一般条款是用于当事人在权利与义务、条款执行条件、资产转移以及条款执行后应满足的后置条件进行约定。若一般条款间在逻辑上具有上下级层次结构, 则称对应下级条款为子条款, 子条款应在特征 **terms** 中进行声明, 子条款标识需通过层次命名方法加以区分。语法如下:

```
<<factor genTerm@termID [#terms={(termID)+}]>>
```

```
text
```

```
<</factor>>
```

其中,

genTerm 一般条款保留字,

termID 条款标识,

terms 子条款关键词。

一般条款按如下四类属性进行标记:

——行为 (**action**): 标记条款的具体动作, **genTerm** 下都需标记条款对应行为属性。语法如下:

```
<<property [termID.]action@actionID #party=(partyID|groupID) #duty=Duty>>
```

```
text
```

```
<</property>>
```

其中,

termID 对应的条款标识,

action 行为保留字,

actionID 行为标识,

party 用于标识条款当事人的特征,

duty 用于标识该当事人权利和义务的特征, 二者在标记中必须明确规定。

权利和义务包括权利限制 (**can**)、应当限制 (**must**) 和禁止限制 (**cannot**), 集合表示如下:

```
Duty ::= {can,must,cannot}
```

——前置条件 (**preCondition**): 标记条款执行前需要满足的要求, 即条款执行条件。语法如下:

```
<<property [termID.]preCondition[@preConditionID]>>
```

```
text
```

```
<</property>>
```

其中，

termID 对应的条款标识，
preCondition 前置条件保留字，
preConditionID termID 条款对应的前置条件标识，超过一处时需添加此标识。

——伴随条件（adjCondition）：标记条款执行过程中伴随的资产操作。语法如下：

```
<<property [termID.]adjCondition[@adjConditionID]>>
```

```
text
```

```
<</property>>
```

其中，

termID 对应的条款标识，
adjCondition 伴随条件保留字，
adjConditionID termID 条款对应的伴随条件标识，超过一处时需添加此标识。

——后置条件（postCondition）：标记条款执行后应满足的条件。语法如下：

```
<<property [termID.]postCondition[@postConditionID]>>
```

```
text
```

```
<</property>>
```

其中，

termID 对应的条款标识，
postCondition 后置条件保留字，
postConditionID termID 条款对应的后置条件标识，超过一处时需添加此标识。

示例：

合同文本：

竞拍者可在竞拍开始后出价竞拍，向合约账户转入大于当前最高价的资金，出价最高者为得主。

标注后文本：

```
<<factor genTerm@term1>>竞拍者
  <<property preCondition>>可在竞拍开始后<</property>>
  <<property action@bid #party=bidder #duty=can>>出价竞拍<</property>>
  <<property adjCondition>>向合约账户转入大于当前最高价的资金<</property>>
  <<property postCondition>>出价最高者为得主<</property>>
<</factor>>
```

10.5.2 违约条款

违约条款是指约定的当事人存在未履行合同中规定的义务或履行义务不符合约定时，违约方应承担的法律责任。当违约条款存在子条款时，也需要在特征 terms 中进行声明。语法如下：

```
<<factor breTerm@breTermID [#terms={{(termID)+}}>>
```

```
text
```

```
<</factor>>
```

其中，

breTerm 违约条款保留字，
breTermID 违约条款标识。

在文本合同中违约条款需要指出本条款针对哪个或哪些条款，CTML 合同使用属性 **against** 对其进行标记，并在特征 **terms** 中对所针对的条款进行声明。语法如下：

```
<<property [breTermID.]against@againstID #terms={{(termID)+}}>>
    text
<</property>>
```

其中，

against 违反保留字，
againstID 违反标识，
terms 特征关键字。

违约条款要素中同样具有 **action**、**preCondition**、**adjCondition** 和 **postCondition** 四类属性，且必须对行为 **action** 进行标注。

注：违约条款可规定违约者必须执行的动作或受害者可以执行的动作。

示例：

合同文本：

若乙方未能够按期付款，则按规定给甲方 5% 滞纳金。

标注后文本：

```
<<factor breTerm@term2>>若乙方
    <<property against@meetPaymentDeadline>>未能够按期付款<</property>>
    <<property action@payExtra #duty=must #party=Buyer>>则按规定给<</property>>
    <<property adjCondition>> 甲方 5% 滞纳金<</property>>
<</factor>>
```

10.5.3 仲裁条款

仲裁条款用于对各种合同争议规定解决争议的方法，并可指定裁决机构。语法如下：

```
<<factor arbiTerm@termID>>
    text
<</factor>>
```

其中，

arbiTerm 仲裁条款保留字，
termID 条款标识。

仲裁条款要素可对合同争议和仲裁机构的陈述进行标注，包括：

——争议（**controversy**）：

```
<<property [termID.]controversy[@controveryID]>>
    text
<</property>>
```

其中，

termID 对应的要素标识，
controversy 争议保留字，
controveryID termID 条款对应的争论标识，超过一处时需添加此标识。

——机构（**institution**）：

```
<<property [termID.]institution[@institutionID]>>
    text
<</property>>
```

其中，

termID	对应的要素标识，
institution	争议保留字，
institutionID	termID 条款对应的机构标识，超过一处时需添加此标识。

示例：

合同文本：

如发生违反租赁条款的争议，双方协商解决；协商不成的，双方愿向北京仲裁委员会申请仲裁。

标注后文本：

```
<<factor arbiTerm@term3>>如发生
  <<property controversy>>违反租赁条款的争议，<</property>>
  双方协商解决；协商不成的，双方愿向
  <<property institution@BAC>>北京仲裁委员会<</property>>
  申请仲裁。
<</factor>>
```

10.6 资产操作

10.6.1 界定和构成

资产操作标注用于实现标的物在不同当事人之间的转移动作提供提取、识别和形式化表示，分为存入动作、取回动作、转移动作三种行为。

10.6.2 存入动作

存入动作标记用于描述当事人主动存入资产。语法如下：

```
<<deposit [#condition=STRING]>>
    text
<</deposit>>
```

其中，

deposit	存入动作保留字，
condition	价值要求，特征值规定为字符串类型（STRING）。

示例 1：

合同文本：

存入 10 元

标注后文本：

```
<<deposit>>存入 10 元<</deposit>>
```

示例 2：

合同文本：

存入大于当前最高价的金额。

标注后文本：

```
<<deposit #condition="value > highestPrice">>
    存入大于当前最高价的金额
<</deposit>>
```

注 1：大于当前最高价“value > highestPrice”为价值要求，其中 highestPrice 表示当前最高价，为货币类型（Money），需在合同中使用域标记进行详细标识。

注 2：存入动作需内嵌目标资产表达式的标记。完整标注后的文本：

```
<<deposit #condition="value > highestPrice">>
    存入大于
    <<assetExpression #assetTarget=highestPrice>>
        当前最高价
    <</assetExpression >>
    的金额
<</deposit>>
```

10.6.3 取回动作

取回动作标记用于描述当事人取回资产。语法如下：

```
<<withdraw>>
    text
<</withdraw>>
```

其中，

withdraw 取回动作保留字。

示例：

合同文本：

取回本金和利息

标注后文本：

```
<<withdraw>>取回本金和利息<</withdraw>>
```

注 1：取回动作需内嵌目标资产表达式的标记。完整标注后的文本：

```
<<withdraw>>
    取回
    <<assetExpression #amount=1+rate #assetTarget=principal>>
        本金和利息
    <</assetExpression>>
<</withdraw>>
```

注 2：本金和利息可表示成“本金*（1+息率）”，其中，rate 表示息率，为浮点类型（Float），principal 表示本金，为货币类型（Money），需在合同中使用域标记进行标识。

10.6.4 转移动作

转移动作标记用于描述向特定目标当事人转移资产。语法如下：

```
<<transfer #target=partyID>>
    text
<</transfer>>
```

其中，

transfer 转移动作保留字，
target 转移目标关键词，指示资产转移目标，
partyID 当事人标识。

示例：

合同文本：

向卖家转移保证金。

标注后文本：

<<transfer #target=seller>>向卖家转移保证金<</transfer>>

注：转移动作需内嵌目标资产表达式的标记。完整标注后的文本：

```
<<transfer #target=seller>>
  向卖家转移
  <<assetExpression #assetTarget=welfare>>
    保证金
  <</assetExpression>>
<</transfer>>
```

10.7 资产表达式

资产表达式用于条款中资产的调用，包括数量、权属、资产标识。语法如下：

```
<<assetExpression [#amount=(FLOAT|INT)] [#assetRight=fieldID] #assetTarget=assetID>>
    text
<</assetExpression>>
```

其中，

assetExpression 资产表达式保留字，
amount 资产数量关键词，对应特征值应为浮点型（FLOAT）或整型（INT），
assetRight 资产权属关键词，对应特征值为标的物权属，
assetTarget 特定资产关键词，对应特征值为资产标识，在声明时不允许被缺省。
 注：amount 在缺省情况下默认特征值为 1，assetRight 在缺省情况下默认合约处置对象的权属为资产所有权。

示例 1：

合同文本：

20 元

标注后文本：

```
<<assetExpression #amount=20 #assetTarget=RMB>>20 元<</assetExpression>>
```

注：资产标识可包括预定义的货币资产，如人民币（RMB）、美元（USD）及货币（Money）类型域

示例 2：

合同文本：

5%滞纳金

标注后文本：

```
<<assetExpression #amount=0.05 #assetTarget=overdueFine>>
  5% 滞纳金
<</assetExpression>>
```

示例 3：

合同文本：

资产房屋的 50%所有权

标注后文本：

```
<<assetExpression #amount=0.5 #assetRight=ownership #assetTarget=House>>
  房屋的 50%所有权
<</assetExpression>>
```

10.8 时间表达式

10.8.1 界定和构成

时间表达式为CTML合同中时间信息提供提取和表示的能力，可分为行为时间、时间谓词、边界谓词。

10.8.2 行为时间

行为时间用于描述当事人完成某项动作的行为时间。语法如下：

```
<<actionTime [#limit=(all|some|this)] #party=(partyID|groupID) #action=actionID>>
    text
<</actionTime>>
```

其中，

actionTime 行为时间保留字，
limit 限制冠词，包括任意（all）、存在（some）或当前（this）三种限制信息，
party 当事人关键词，其特征值为个体当事人标识（partyID）或群体当事人（groupID），
action 行为关键词，
actionID 行为标识。

注：根据当事人为个体或群体情况，可将行为时间表示分为如下两种情况：

——当事人为个体时，不必添加冠词 **all**、**some**、**this**。

示例 1：

合同文本：

买家支付完成

标注后文本：

```
<<actionTime #party=buyer #action=pay>>买家支付完成<</actionTime>>
```

——当事人为群体时，可通过添加冠词 **all**、**some**、**this** 表示特定时间。

- 冠词**all**：表示群体当事人中最后一个完成某动作的时间。

示例 2：

合同文本：

所有投票者投票完成

标注后文本：

```
<<actionTime #limit=all #party=voters #action=vote>>
```

所有投票者投票完成

```
<</actionTime>>
```

- 冠词**some**：表示群体当事人最近一个完成某动作的时间。

示例 3：

合同文本：

竞标者群体中最近一个完成拍卖

标注后文本：

```
<<actionTime #limit=some #party=bidder #action=bid>>
```

竞标者群体中最近一个完成拍卖

```
<</actionTime>>
```

- 冠词**this**：当前条款执行人若属于该群体，则表示该个体的完成时间。

示例 4：

合同文本：

本人投票完成

标注后文本：

```
<<actionTime #limit=this #party=voter #action=vote>>
  本人投票完成
<</actionTime>>
```

10.8.3 时间谓词

时间谓词用于描述目标时间与基准时间之间的关系。语法如下：

```
<<timePredicate [#targetTime=timeID] [#judge=(is|isn't)]
  [#prep=(before|after)] #baseTime=timePoint>>
  text
<</timePredicate>>
```

其中，

timePredicate	时间谓词保留字，
targetTime	目标时间特征关键词，
timeID	指定时间标识，
judge	为判断词关键词，特征值可选择是（is）否（isn't），
prep	为时间方向介词，特征值可选择时间之前（before）和时间之后（after），
baseTime	为基准时间关键词，
timePoint	包括时间类型（Date），当前时间（now）和行为时间（action），即

timepoint::={Date,now,actionTime}

示例 1：

合同文本：

在截止日期之前

标注后文本：

```
<<timePredicate #prep=before #baseTime=expirationDate>>在截止日期之前<</timePredicate>>
```

示例 2：

合同文本：

本人投票完成时刻是在生效日期之后

标注后文本：

```
<<timePredicate #targetTime=voteEnd #judge=is #prep=after #baseTime=effectiveDate>>
  本人投票完成时刻是在生效日期之后
<</timePredicate>>
```

注：voteEnd 指当事人完成投票动作的时间，effectiveDate 指生效日期。

10.8.4 边界谓词

边界谓词用于描述某一基准时间前后的某一范围，语法如下：

```
<<rangePredicate [#within=(true|false)] #boundary=TIME
  [#prep=(before|after)] #baseTime=timePoint>>
  text
<</rangePredicate>>
```

其中，

rangePredicate	边界谓词保留字，
within	边界判定关键词，特征值为布尔值，
boundary	时间范围关键词，特征值为时间类型（TIME）。

示例 1:

合同文本:

在拍卖结束前的三天内。

标注后文本:

```
<<rangePredicate #within=true #boundary=3day #prep=before #baseTime=auctionEnd>>
    在拍卖结束前的三天内
<</rangePredicate>>
```

注: auctionEnd 指拍卖结束的日期。

示例 2:

合同文本:

在拍卖结束前的三天以前。

标注后文本:

```
<<rangePredicate #boundary=3day #prep=before #baseTime=auctionEnd>>
    在拍卖结束前的三天以前
<</rangePredicate>>
```

10.9 附加信息

附加信息标记用于描述其它补充条件。语法如下:

```
<<factor addition@additionID >>
    text
<</factor>>
```

其中,

addition 附加信息保留字,

additionID 附加信息标识。

注: 附加信息中应使用域对其拥有的特性及其取值进行标注。

示例 1:

合同文本:

合同信息如保证金等

标注后文本:

```
<<factor addition@contractInfo>>合同信息如
    <<field contractInfo@welfare>>
        保证金
    <</field>>等
<</factor>>
```

示例 2:

合同文本:

最高出价

标注后文本:

```
<<field @highestPrice>>最高出价<</field>>
```

10.10 合约订立

合同订立标记用于描述所有当事人的约定总结, 表明缔约当事人相互为意思表示, 达成合意, 同意签订合同。语法如下:

<<factor conclusion@conclusionID>>

text

<</factor>>

其中，

conclusion 合同订立保留字，

conclusionID 合同订立标识。

合同订立要素应包含当事人的签名，代表各方同意合同相关陈述并对合同进行签署。语法如下：

<<property [conclusionID.]signature[@signatureID] #party=(partyID|groupID)>>

text

<</property>>

其中，

conclusionID 当前合同订立标识，

signature 签名保留字，

signatureID 签名标识，

party 特征关键词，其特征值为签名对应的当事人标识。

示例：

<<factor conclusion@SellContractConclusion>>

本智能法律合约当事人不得以任何形式修改本合约，除非以书面形式并经双方签字。本合约及其附件构成合约双方的完整协议。本合约对当事各方及其继承人、受让人均具有约束力。通过签署本协议，各方同意上述条款。双方各持到一份本协议，并负责维护其条款。双方同意将本合约转化为智能合约平台上的计算机程序，并同意该程序及其执行具有相同法律效力。

<<property signature #party=Seller>>

甲方（签名或盖章）：

签订日期：

<</property>>

<<property signature #party=Buyer>>

乙方（签名或盖章）：

签订日期：

<</property>>

<</factor>>

附 录 A
(资料性)
法律文本合同及标注后 CTML 合同示例

A.1 法律文本合同

房屋买卖合同

卖方：_____（以下简称甲方）

买方：_____（以下简称乙方）

（一）、为房屋买卖有关事宜，经双方协商，订合同如下：甲方自愿下列房屋卖给乙方所有：

1. 房屋状况：（请按《房屋所有权证》填写）

房屋坐落 幢号 室号 套（间）数 建筑结构 总层数 建筑面积（平方） 用途

2. 该房屋的土地使用权取得方式“√”：出让（ ）划拨（ ）

（二）、甲乙双方商定成交价格为人民币_____元，（大写）_____佰_____拾_____万_____仟_____佰_____拾 元整。

乙方在_____年_____月_____日前分_____次付清，付款方式：

（三）、甲方在_____年_____月_____日将上述房屋交付给乙方。该房屋占用范围内土地使用权同时转让。

（四）、若乙方未能够按期付款，则按规定给甲方 5%滞纳金，滞纳金根据民法典规定予以得出。

（五）、本合同经双方签章并经嘉兴市房地产交易管理所审查鉴定后生效，并对双方都具有约束力，应严格履行。如有违约，违约方愿承担违约责任，并赔偿损失，支付违约金。

（六）、双方愿按国家规定交纳税、费及办理有关手续。未尽事宜，双方愿按国家有关规定办理。

如发生争议，双方协商解决；协商不成的，双方愿向（_____）仲裁委员会申请仲裁。

（七）、本合同一式四份，甲、乙双方及税务部门各一份，房管部门一份。

（八）、双方约定的其它事项：_____

甲方（签名或盖章）_____

签订日期：_____年_____月_____日

乙方（签名或盖章）_____

签订日期：_____年_____月_____日

A.2 标注后 CTML 合同

```

<!DOCTYPE ctml>
<ctml lang=zh_cn, link="exchangesheet.emd">
  <<factor title@printerDeal>> 房屋买卖合同 <</factor>>
  <<factor party@Seller>>卖方: <{Seller@name %String}> (以下简称甲方) <</factor>>
  <<factor party@Buyer>>买方: <{Buyer@name %String}> (以下简称乙方) <</factor>>
  (一)、为房屋买卖有关事宜, 经双方协商, 订合同如下: 甲方自愿下列房屋卖给乙方所有:
  <<factor asset@House>>
    <<property info>>1.房屋状况: (请按《房屋所有权证》填写)
      <<field @location>>房屋坐落 <{House@location}> <</field>>
      <<field @blockNumber>>幢号 <{House@blockNumber}> <</field>>
      <<field @roomNumber>>室号 <{House@roomNumber}> <</field>>
      <<field @units>>套(间)数 <{House@units %Integer}> <</field>>
      <<field @structure>>建筑结构 <{House@structure}> <</field>>
      <<field @floor>>总层数 <{House@floor %Integer}> <</field>>
      <<field @area>>建筑面积(平方) <{House@area %Integer}> <</field>>
      <<field @usage>>用途 <{House@usage}> <</field>>
    <</property>>
    2.<<property right>>该房屋的
      <<field @houseLandUse %Right>>土地使用权取得方式“√”:
        出让 (<{@accessMethod %Right #singleOption={transferred}}>)
        划拨 (<{@accessMethod %Right #singleOption={allocation}}>)
      <</field>>
    <</property>>
  <</factor>>
  <<property term1.adjCondition>>
    (二)、甲乙双方商定
    <<transfer #target=Seller>>成交价格为
      <<assetExpression #assetTarget=RMB>>人民币
      <<field @dealPrice %Money>>
        <{@dealPrice %Money}>元
      <</field>>
      <</assetExpression>>, (大写) <{@dealPrice %Money}> 元整。
    <</transfer>>
  <</property>>
  <<factor genTerm@term1>>
    乙方

```

<<property preCondition>>在
 <<timePredicate #prep=before #baseTime=expirationDate>>
 <<field @expirationDate %Date>>
 <{@expirationDate %Date}> 日前
 <</field>>
 <</timePredicate>> 分 <{term1 @frequency %Integer}> 次
<</property>>
<<property action@afford #duty=must #party=Buyer>> 付清 <</property>>,
付款方式: <{term1 @paymentMethod}>
<</factor>>
<<factor genTerm@term2>>
(三)、甲方
 <<property preCondition>>
 <<rangePredicate #within=true #boundary=1day #prep=after
#baseTime=houseDeadline>>在
 <<field @houseDeadline %Date>>
 <{term2@houseDeadline %Date}>
 <</field>>
 <</rangePredicate>>
 <</property>>
 <<property adjCondition@adj1>>
 <<transfer #target=Buyer>>将
 <<assetExpression #assetTarget=House>>
 上述房屋
 <</assetExpression>>
 <</transfer>>
 <</property>>
 <<property action@deliver #party=Seller #duty=must>>
 交付
 <</property>>给乙方。
 <<property adjCondition@adj2>>该房屋
 <<transfer #target=Buyer>>占用范围内
 <<assetExpression #assetRight=houseLandUse #assetTarget=House>>
 土地使用权
 <</assetExpression>>同时转让
 <</transfer>>
 <</property>>。
<</factor>>

<<factor breTerm@term3>>

(四)、若乙方

<<property against@meetPaymentDeadline #terms={term1}>>

未能够按期付款

<</property>>, 则

<<property action@payExtra #duty=must #party=Buyer>>

按规定给

<</property>>

<<property adjCondition>>

<<transfer #target=Seller>> 甲方

<<assetExpression #amount=0.05 #assetTarget=contractInfo.overdueFine >>

5% 滞纳金

<</assetExpression>>

<</transfer>>

<</property>>

<</factor>>,

<<factor addition@contractInfo>>

<<field contractInfo@overdueFine>>

滞纳金

<</field>>

<<factor>>根据民法典规定予以得出。

<<factor conclusion@Conclusion1>>

(五)、本合同经双方签章并经嘉兴市房地产交易管理所审查鉴定后生效,并对双方都具有约束力,应严格履行。如有违约,违约方愿承担违约责任,并赔偿损失,支付违约费用。

<</factor>>

<<factor conclusion@Conclusion2>>

(六)、双方愿按国家规定交纳税、费及办理有关手续。未尽事宜,双方愿按国家有关规定办理。

<</factor>>

<<factor arbiTerm@term4>>

如发生争议,双方协商解决;协商不成的,双方愿向

<<property institution>> (<{term4@committee %String}>) 仲裁委员会

<</property>>申请仲裁。

<</factor>>

<<factor conclusion@Conclusion3>>

(七)、本合同一式四份,甲、乙双方及税务部门各一份,房管部门一份。

(八)、双方约定的其它事项: <{Conclusion3@promise %String}>

<<property signature #party=Seller>>甲方(签名或盖章)

<<field Seller@SellerSignature>>

<{Seller@SellerSignature}>

```
<</field>>签订日期:
<<field Seller@SellerSignatureDate>>
    <{Seller@SellerSignatureDate %Date}>
<</field>>
<</property>>
<<property signature #party=Buyer>>乙方（签名或盖章）
    <<field Buyer@BuyerSignature>>
        <{Buyer@BuyerSignature}>
    <</field>>签订日期:
    <<field Buyer@BuyerSignatureDate>>
        <{Buyer@BuyerSignatureDate %Date}>
    <</field>>
<</property>>
<</factor>>
</ctl>
```

附录 B (资料性) EMD 的交互数据属性与示例

B.1 EMD 的交互数据属性

交互数据属性是指合同协商与执行过程中对当事人交互数据进行约束与限定的特定性质或关系。交互数据属性可包括：

- 使用方法usage：用于对交互数据的操作状态进行约束，可使用六个字符进行表示，每两个字符为一组，分别为合约签订前、中、后阶段对变量的状态权限约束，逻辑关系包括：
 - R表示此变量仅可读（Readable）；
 - W表示此变量可写（Writeable）；
 - C表示此变量需在此阶段被确认（Confirmed）；
 - U表示此变量无需在此阶段被确认（Unconfirmed）；
 - C与W共同使用形成WC，表示此变量必须填写；
 - U和W共同使用形成WU，表示此变量非必须填写；
 - 此变量为可读状态时，其权限约束表示用“_”表示。
- 来源标识ctmlID：交互数据在CTML合同中的对应标识；
- 目标标识targetID：交互数据在被转化后智能合约中所对应变量名；
- 类型type：交互数据的数据类型；
- 缺省值defaultValue：交互数据的缺省值；
- 确值条件condition：用于对交互数据的值域范围限定、填写人约束条件进行约束：
 - 单选singleOption：表明相应交互数据为单选项，并记录取值范围；
 - 多选multiOption：表明相应交互数据为多选项，并记录取值范围；
 - 当事人标识partyID：表明相应交互数据需由该标识对应当事人填写；
- 交互数据值value：交互过程中客户填写的交互数据数值。

B.2 EMD 示例

EMD生成流程如下：

- a) 逐个提取数源标记，生成一个以交互数据标识命名的记录；
- b) 客户应通过选择或编辑方式对记录中的交互数据属性进行确认、特指或限制；
- c) 合约服务器应将上述用户指定的交互数据属性写入到交换标记数据表EMD对应记录中。

根据上述流程，附录A中标注后的CTML合同为可得到EMD如下表所示：

表 2 交换数据表示例

Factor	ID	Usage	ctmlID	Value	TargetID	Type	Default-Value	Condition
Seller	name	WCWUR_	name	\	name	String	\	must Seller
	SellerSign- ature	WCWUR_	SellerSign- ature	\	SellerSign- ature	String	\	must Seller
	SellerSign- atureDate	WCWUR_	SellerSign- atureDate	\	SellerSign- ature	Date	\	must System
Buyer	name	WCWUR_	name	\	name	String	\	must Buyer
	Buyer -Signature	WCWUR_	Buyer -Signature	\	Buyer -Signature	String	\	must Buyer
	BuyerSign- atureDate	WCWUR_	BuyerSign- atureDate	\	BuyerSign- atureDate	Date	\	must System
House	location	WUR_R_	location	\	location	String	\	\
	blockNumber	WUR_R_	blockNumb er	value	blockNumber	String	\	\
	accessMethod	WUWCR_	accessMe- thod	value	accessMethod	String	transferred	singleOptio n,[transferre d, allocation]
term1	afford -Deadline	WUWCR_	afford -Deadline	\	afford -Deadline	Date	\	\
	frequency	WUWCR_	frequency	\	frequency	Intege r	\	\
	payment -Method	WUWCR_	payment -Method	\	payment -Method	String	\	\
	dealPrice	WUWCR_	dealPrice	\	dealPrice	Mone y	\	\
term2	houseDead- line	WUWCR_	houseDead- line	\	houseDead- line	Date	\	\
term4	committee	WUWCR_	committee	\	committee	String	\	\
Conclu- sion3	promise	WUWCR_	promise	\	promise	String	\	\

附录 C
(资料性)
CTML 到 SLCL 转化关系表

CTML到SLCL的转化关系表现为由CTML中标记到SLCL中符号的元素映射关系，如表3所述。

元素映射关系见表3中第四列，采用符号“:”表示映射关系，即“CTML中的标记”:“SLCL中的符号”。

表 3 CTML 到 SLCL 转化关系表

要素	CTML	SLCL	元素映射关系
合约名称	<<factor title@titleID [#serialNumber=value]>> text <</factor>>	<i>Title</i> ::= contract Cname (: serial number Chash)?	titleID: Cname value: Chash
当事人描述	<<factor party@partyID>> text <</factor>>	<i>Parties</i> ::= party group ? Pname { <i>field</i> + }	partyID: Pname
	<<factor group@groupID>> text <</factor>>		groupID: Pname
	<<field [factorID.propertySet]@fieldID [%type] [#quantity=value]>> text <</field>>	<i>field</i> ::= attribute : (constant type)	fieldID: attribute value: constant type: type
标的	<<factor asset@assetID>> text <</factor>>	<i>Assets</i> ::= asset Aname{ info { <i>field</i> + } right { <i>field</i> + } }	assetID: Aname
	<<property [assetID.]info>> text <</property>>		info: info { }
	<<property [assetID.]right>> text <</property>>		right: right { }
	<<field [factorID.propertySet]@fieldID [%type] [#quantity=value]>> text <</field>>	<i>field</i> ::= attribute : (constant type)	fieldID: attribute value: constant type: type
	<<field [factorID.propertySet]@fieldID [%rightType] [#quantity=value]>> text <</field>>		fieldID: attribute value: constant rightType: type
一般条款	<<factor genTerm@termID [#terms={ (termID)+ }]>> text <</factor>>	<i>GeneralTerms</i> ::= term Tname:	termID: Tname
	<<property [termID.]action@actionID #party=(partyID groupID) #duty=Duty>> text <</property>>	Pname (must can cannot) action(<i>field</i> +)	actionID: action Duty: (must can cannot) (partyID groupID): Pname

	<<property [termID.]preCondition[@preConditionID]>> text <</property>>	(when preCondition)?	preCondition: when
	<<property [termID.]adjCondition[@adjConditionID]>> text <</property>>	(while transactions+)?	adjCondition: while
	<<property [termID.]postCondition[@postConditionID]> > text <</property>>	(where postCondition)?.	postCondition: where
违 约 条 款	<<factor breTerm@breTermID [#terms={ (termID)+ }]>> text <</factor>>	BreachTerms ::= breach term Bname	breTermID: Bname
	<<property [breTermID.]against@againstID #terms={ (termID)+ }>> text <</property>>	(against Tname+)?	termID: Tname
	<<property [termID.]action@actionID #party=(partyID groupID) #duty=Duty>> text <</property>>	Pname (must can) action(field+)	actionID: action Duty: (must can) (partyID groupID): Pname
	<<property [termID.]preCondition[@preConditionID]>> text <</property>>	(when preCondition)?	preCondition: when
	<<property [termID.]adjCondition[@adjConditionID]>> text <</property>>	(while transactions+)?	adjCondition: while
	<<property [termID.]postCondition[@postConditionID]> > text <</property>>	(where postCondition)?.	postCondition: where
仲 裁 条 款	<<factor arbiTerm@termID>> text <</factor>>	ArbitrationTerms ::= arbitration term : (The statement of any controversy)?	
	<<property [termID.]controversy[@controversyID]>> text <</property>>		text: (The statement of any controversy)
	<<property [termID.]institution[@institutionID]>> text <</property>>	administered by institution : instName	institutionID: instName
附 加 信 息	<<field [factorID.propertySet]@fieldID [%type] [#quantity=value]>> text <</field>>	Additions ::= field + (addition Dname { field + }) <i>field</i> ::= attribute : (constant type)	fieldID: attribute value: constant type: type
	<<factor addition@additionID >> text <</factor>>		additionID: Dname
合 同	<<factor conclusion@conclusionID>> text <</factor>>	Signs ::= Contract conclusions : (The statement of all parties.)?	text: (The statement of all parties.)

订 立	<<property [conclusionID.]signature [@signatureID] #party= (partyID groupID) >> text <</property>>	Signature of party Pname : {printedName: String, signature: String, date: Date }	(partyID groupID): Pname
资 产 表 示	<<assetExpression [#amount=(FLOAT INT)] [#assetRight=fieldID] #assetTarget=assetID>> text <</assetExpression>>	AssetExpression ::= \$ (amount)? (right of)? Aname	(FLOAT INT): (amount) fieldID: right assetID: Aname
资 产 操 作	<<deposit [#condition=STRING]>> text <</deposit>>	Deposits ::= deposit (value RelationOperator)? AssetExpression	STRING: (value RelationOperator)
	<<withdraw>> text <</withdraw>>	Withdraws ::= withdraw AssetExpression	
	<<transfer #target=partyID>> text <</transfer>>	Transfers ::= transfer AssetExpression to target	partyID: target
时 间 表 示	<<actionTime [#limit=(all some this)] #party=(partyID groupID) #action=actionID>> text <</actionTime>>	ActionEnforcedTimeQuery ::= (all some this)? party did action	(all some this): (all some this) (partyID groupID): party actionID: action
	<<timePredicate [#targetTime=timeID] [#judge=(is isn't)] [#prep=(before after)] #baseTime=timePoint>> text <</timePredicate>>	TimePredicate ::= (targetTime)? (is isn't) (before after) baseTime	timeID: (targetTime) (is isn't): (is isn't) (before after): (before after) timePoint: baseTime
	<<rangePredicate [#within=(true false)] #boundary=TIME [#prep=(before after)] #baseTime=timePoint>> text <</rangePredicate>>	BoundedTimePredicate ::= (within)? boundary (before after) baseTime	(true false): (within)? TIME: boundary (before after): (before after) timePoint: baseTime

参考文献

- [1] GB/T 35285-2017 信息安全技术 公钥基础设施 基于数字证书的可靠电子签名生成及验证技术要求
 - [2] YD/T 3204-2016 网络电子身份标识 eID 体系架构
 - [3] 中华人民共和国民法典
 - [4] 中华人民共和国电子签名法
-

